Here is an example on how to use MAPLE to do all the calculations in Homework 6. The following is problem 4.6.23

```
> restart;with(DEtools):with(linalg):with(plots):
> de:=diff(w(t),t$2) - 4*diff(w(t),t) +2*w(t) = 0;
```

$$de := (\frac{\partial^2}{\partial t^2}\,\mathrm{w}(t)) - 4\,(\frac{\partial}{\partial t}\,\mathrm{w}(t)) + 2\,\mathrm{w}(t) = 0$$

Now, substitute the usual expression $w = e^{rt}$

```
> eval(subs(w(t)=exp(r*t),lhs(de)));
```

$$r^2\,e^{(r\,t)} - 4\,r\,e^{(r\,t)} + 2\,e^{(r\,t)}$$

We cancel the exponential factor to arrive at the characteristic polynomial

```
> simplify(%/exp(r*t));
```

$$r^2 - 4\,r + 2$$

Now convert the axpression above to a function of r in order to solve for the roots of p(r)=0:

```
> poly:=unapply(%,r);
```

$$poly := r \to r^2 - 4\,r + 2$$

Use SOLVE to find the roots of the characteristic polynomial

```
> solve(poly(r),r);
```

$$2 + \sqrt{2},\ 2 - \sqrt{2}$$

2 real roots! The easier case: define a function so we can solve the IVP:

```
> y_h:=t->C1*exp((2+2^(1/2))*t)+C2*exp((2-2^(1/2))*t);
```

$$y\_h := t \to C1\,e^{((2+\sqrt{2})\,t)} + C2\,e^{((2-\sqrt{2})\,t)}$$

Now compute the t-derivative:

```
> yh_p:=diff(y_h(t),t);
```

$$yh\_p := C1\,(2+\sqrt{2})\,e^{((2+\sqrt{2})\,t)} + C2\,(2-\sqrt{2})\,e^{((2-\sqrt{2})\,t)}$$

Now evaluate the t-derivative at t=0, and set the resulting expression equal to the corresponding initial value:

```
> yp_0:=eval(subs(t=0,yh_p))=1;
```

$$yp\_0 := C1\,(2+\sqrt{2}) + C2\,(2-\sqrt{2}) = 1$$

Do the same for the value of y(0):

```
> y_0:=y_h(0)=0;
```

$$y\_0 := C1 + C2 = 0$$

```
Now solve the resulting system for C1 and C2:
> solve({y_0,yp_0},{C1,C2});
```

$$\{C1 = \frac{1}{4}\sqrt{2},\ C2 = -\frac{1}{4}\sqrt{2}\}$$

```
Having found the values of C1, C2, substitute into the homogeneous
solution to find desired function
```

```
> y_h:=t->sqrt(2)/4*exp((2+2^(1/2))*t)-sqrt(2)/4*exp((2-2^(1/2))*t);
```

$$y\_h := t \to \frac{1}{4}\sqrt{2}\, e^{((2+\sqrt{2})\, t)} - \frac{1}{4}\sqrt{2}\, e^{((2-\sqrt{2})\, t)}$$

```
> plot(y_h,-2..2,axes=FRAME);
```